
Gram�aticas libres de contexto

x

Ejemplos de frases

u

El gato come pescado

u

El perro come carne

x

Ejemplo de gram�atica

oraci�on --> sintagma_nominal, sintagma_verbal

sintagma_nominal --> nombre

sintagma_nominal --> art��culo, nombre

sintagma_verbal --> verbo, sintagma_nominal

art��culo --> [el]

nombre --> [gato]

nombre --> [perro]

nombre --> [pescado]

nombre --> [carne]

verbo --> [come]

x

Arbol de an�alisis

oraci�on

|

+-------+-------+

| |

sintagma_nominal sintagma_verbal

| |

+---+---+ +---+---+

| | | |

art��culo nombre verbo nombre

| | | |

el gato come pescado

Gram�aticas libres de contexto

x

Concepto de gram�atica: G = (N,T,P,S)

u

N: vocabulario no terminal (categor��as sint�acticas)

u

T: vocabulario terminal

u

P: reglas de producci�on

u

S: s��mbolo inicial

x

Vocabulario

u

V = N [ T es el vocabulario

u

N \ T = ;

x

Gram�aticas libres de contextos

A =) w, con A 2 N y w 2 V

�

x

Derivaciones

u

xAy =) xwy mediante A =) w

u

x

�

=) y si existen x

1

; x

2

; : : : ; x

n

tales que

x = x

1

=) x

2

� � � =) x

n�1

=) x

n

= y

x

Lenguaje de�nido por una gram�atica

L(G) = fx 2 T

�

: S

�

=) xg

Gram�aticas libres de contexto en Prolog

x

Representaci�on de oraciones en Prolog

[el, gato, come, pescado]

[el, perro, come, carne]

x

Gram�atica en Prolog con listas

u

Sesi�on

?- oracion([el,gato,come,pescado]).

Yes

?- oracion([el,come,pescado]).

No

u

De�nici�on

oracion(O) :- sintagma_nominal(SN),

sintagma_verbal(SV),

append(SN,SV,O).

sintagma_nominal(SN) :- nombre(SN).

sintagma_nominal(SN) :- art��culo(A),

nombre(N),

append(A,N,SN).

sintagma_verbal(SV) :- verbo(V),

sintagma_nominal(SN),

append(V,SN,SV).

art��culo([el]).

nombre([gato]).

nombre([perro]).

nombre([pescado]).

nombre([carne]).

verbo([come]).

Gram�aticas libres de contexto en Prolog

x

Gram�atica en Prolog con listas de diferencia

u

Sesi�on

?- oracion([el,gato,come,pescado]-[]).

Yes

?- oracion([el,come,pescado]-[]).

No

u

De�nici�on

oracion(L-L0) :-

sintagma_nominal(L-L1),

sintagma_verbal(L1-L0).

sintagma_nominal(L-L0) :-

nombre(L-L0).

sintagma_nominal(L-L0) :-

art��culo(L-L1),

nombre(L1-L0).

sintagma_verbal(L-L0) :-

verbo(L-L1),

sintagma_nominal(L1-L0).

art��culo([el|L]-L).

nombre([gato|L]-L).

nombre([perro|L]-L).

nombre([pescado|L]-L).

nombre([carne|L]-L).

verbo([come|L]-L).

Gram�aticas de cl�ausulas de�nidas

x

Ejemplo de GCD

u

De�nici�on

oraci�on --> sintagma_nominal,

sintagma_verbal.

sintagma_nominal --> nombre.

sintagma_nominal --> art��culo,

nombre.

sintagma_verbal --> verbo,

sintagma_nominal.

art��culo --> [el].

nombre --> [gato].

nombre --> [perro].

nombre --> [pescado].

nombre --> [carne].

verbo --> [come].

Gram�aticas de cl�ausulas de�nidas

u

Sesi�on

?- listing(oraci�on).

oraci�on(A, B) :-

sintagma_nominal(A, C),

sintagma_verbal(C, B).

Yes

?- phrase(oraci�on,[el,gato,come,pescado]).

Yes

?- phrase(oraci�on,[el,come,pescado]).

No

?- phrase(oraci�on,[el,gato,X,pescado]).

X = come ;

No

?- phrase(oraci�on,[X,gato,Y,pescado]).

X = el

Y = come ;

No

?- phrase(sintagma_nominal,L).

L = [gato] ;

L = [perro]

Yes

Gram�aticas de cl�ausulas de�nidas

x

Arbol de an�alisis con GCD

u

Sesi�on

?- phrase(oraci�on(T),[el,gato,come,pescado]).

T = o(sn(art(el), n(gato)), sv(v(come), sn(n(pescado))))

Yes

?- ['../tema-06/p1-arbol-termino.pl'].

Yes

?- phrase(oraci�on(_T),[el,gato,come,pescado]),

termino_escribe(_T).

---------o--------sn-------art--------el

---------n------gato

--------sv---------v------come

--------sn---------n---pescado

Yes

u

De�nici�on

oraci�on(o(SN,SV)) --> sintagma_nominal(SN),

sintagma_verbal(SV).

sintagma_nominal(sn(N)) --> nombre(N).

sintagma_nominal(sn(Art,N)) --> art��culo(Art),

nombre(N).

sintagma_verbal(sv(V,SN)) --> verbo(V),

sintagma_nominal(SN).

art��culo(art(el)) --> [el].

nombre(n(gato)) --> [gato].

nombre(n(perro)) --> [perro].

nombre(n(pescado)) --> [pescado].

nombre(n(carne)) --> [carne].

verbo(v(come)) --> [come].

Gram�aticas de cl�ausulas de�nidas

x

Concordancia de g�enero

u

Sesi�on

?- phrase(oraci�on,[el,gato,come,pescado]).

Yes

?- phrase(oraci�on,[la,gato,come,pescado]).

No

?- phrase(oraci�on,[la,gata,come,pescado]).

Yes

u

De�nici�on

oraci�on --> sintagma_nominal,

sintagma_verbal.

sintagma_nominal --> nombre(_).

sintagma_nominal --> art��culo(G),

nombre(G).

sintagma_verbal --> verbo,

sintagma_nominal.

art��culo(masculino) --> [el].

art��culo(femenino) --> [la].

nombre(masculino) --> [gato].

nombre(femenino) --> [gata].

nombre(masculino) --> [pescado].

verbo --> [come].

Gram�aticas de cl�ausulas de�nidas

x

Concordancia en n�umero

u

Sesi�on

?- phrase(oraci�on,[el,gato,come,pescado]).

Yes

?- phrase(oraci�on,[los,gato,come,pescado]).

No

?- phrase(oraci�on,[los,gatos,comen,pescado]).

Yes

u

De�nici�on

oraci�on --> sintagma_nominal(N),

sintagma_verbal(N).

sintagma_nominal(N) --> nombre(N).

sintagma_nominal(N) --> art��culo(N),

nombre(N).

sintagma_verbal(N) --> verbo(N),

sintagma_nominal(_).

art��culo(singular) --> [el].

art��culo(plural) --> [los].

nombre(singular) --> [gato].

nombre(plural) --> [gatos].

nombre(singular) --> [perro].

nombre(plural) --> [perros].

nombre(singular) --> [pescado].

nombre(singular) --> [carne].

verbo(singular) --> [come].

verbo(plural) --> [comen].

Gram�aticas de cl�ausulas de�nidas

x

GCD con llamadas a Prolog

u

Sesi�on

?- phrase(oraci�on,[el,gato,come,pescado]).

Yes

?- phrase(oraci�on,[los,gato,come,pescado]).

No

?- phrase(oraci�on,[los,gatos,comen,pescado]).

Yes

u

De�nici�on

oraci�on --> sintagma_nominal(N),

sintagma_verbal(N).

sintagma_nominal(N) --> nombre(N).

sintagma_nominal(N) --> art��culo(N),

nombre(N).

sintagma_verbal(N) --> verbo(N),

sintagma_nominal(_).

art��culo(singular) --> [el].

art��culo(plural) --> [los].

nombre(singular) --> [Palabra], {nombre(Palabra, _)}.

nombre(plural) --> [Palabra], {nombre(_, Palabra)}.

verbo(singular) --> [come].

verbo(plural) --> [comen].

nombre(gato, gatos).

nombre(perro, perros).

nombre(pescado, pescados).

nombre(carne, carnes).

Gram�aticas de cl�ausulas de�nidas

x

Concordancia en g�enero y n�umero

u

Sesi�on

?- phrase(oraci�on,[la,profesora,lee,un,libro]).

Yes

?- phrase(oraci�on,[la,profesor,lee,un,libro]).

No

?- phrase(oraci�on,[los,profesores,leen,un,libro]).

Yes

?- phrase(oraci�on,[los,profesores,leen]).

Yes

?- phrase(oraci�on,[los,profesores,leen,libros]).

Yes

Gram�aticas de cl�ausulas de�nidas

u

De�nici�on

oraci�on --> sintagma_nominal(N),

verbo_(N),

complemento.

complemento --> [].

complemento --> sintagma_nominal(_).

sintagma_nominal(N) --> nombre_(_,N).

sintagma_nominal(N) --> determinante_(G,N),

nombre_(G,N).

verbo_(N) --> [P],{verbo(P,N)}.

nombre_(G,N) --> [P],{nombre(P,G,N)}.

determinante_(G,N) --> [P],{determinante(P,G,N)}.

nombre(profesor,masculino,singular).

nombre(profesores,masculino,plural).

nombre(profesora,femenino,singular).

nombre(profesoras,femenino,plural).

nombre(libro,masculino,singular).

nombre(libros,masculino,plural).

determinante(el,masculino,singular).

determinante(los,masculino,plural).

determinante(la,femenino,singular).

determinante(las,femenino,plural).

determinante(un,masculino,singular).

determinante(una,femenino,singular).

determinante(unos,masculino,plural).

determinante(unas,femenino,plural).

verbo(lee,singular).

verbo(leen,plural).

Sem�antica de expresiones aritm�eticas

x

Sesi�on

?- phrase(exp_numerica(N),['(',2,+,3,')']).

N = 5 ;

No

?- phrase(exp_numerica(N),['(','(',2,+,3,')',*,4,')']).

N = 20 ;

No

x

De�nici�on

exp_numerica(N) --> [N], {integer(N)}.

exp_numerica(N) --> ['('],

exp_numerica(N),

[')'].

exp_numerica(N) --> ['('],

exp_numerica(N1),

operacion(Op),

exp_numerica(N2),

[')'],

{aplica(Op,N1,N2,N)}.

operacion(+) --> [+].

operacion(-) --> [-].

operacion(*) --> [*].

operacion(//) --> [/].

aplica(Op,N1,N2,N) :-

T =.. [Op,N1,N2],

N is T.

Razonamiento con lenguaje natural

x

Gram�atica de asertos y preguntas

u

Ejemplos

?- phrase(oraci�on(O),L).

O = europeo(juan) :- true

L = [juan, es, europeo] ;

O = andaluz(juan) :- true

L = [juan, es, andaluz] ;

O = europeo(_G273) :- europeo(_G273)

L = [todo, europeo, es, europeo] ;

O = andaluz(_G273) :- europeo(_G273)

L = [todo, europeo, es, andaluz] ;

O = europeo(_G273) :- andaluz(_G273)

L = [todo, andaluz, es, europeo] ;

O = andaluz(_G273) :- andaluz(_G273)

L = [todo, andaluz, es, andaluz] ;

No

?- phrase(pregunta(P),L).

P = europeo(juan)

L = [�, es, juan, europeo, ?] ;

P = andaluz(juan)

L = [�, es, juan, andaluz, ?] ;

P = europeo(_G297)

L = [�, qui�en, es, europeo, ?] ;

P = andaluz(_G297)

L = [�, qui�en, es, andaluz, ?] ;

No

Razonamiento con lenguaje natural

u

De�nici�on

:- op(600,xfy, '=>').

oraci�on((L:-true)) --> nombre_propio(X),

sintagma_verbal(X=>L).

oraci�on(C) --> determinante(A1,A2,C),

adjetivo(A1),

sintagma_verbal(A2).

sintagma_verbal(A) --> verbo, adjetivo(A).

pregunta(P) --> [�,es],

nombre_propio(X),

adjetivo(X=>P),

[?].

pregunta(P) --> [�,qui�en,es],

adjetivo(_X=>P),

[?].

nombre_propio(juan) --> [juan].

determinante(X=>Cu,X=>Ca,(Ca:-Cu)) --> [todo].

verbo --> [es].

adjetivo(X=>europeo(X)) --> [europeo].

adjetivo(X=>andaluz(X)) --> [andaluz].

Razonamiento con lenguaje natural

x

Sistema de consulta y razonamiento

u

Ejemplo

?- consulta([]).

? [juan,es,andaluz].

? [�, qui�en, es, andaluz, ?].

! [juan, es, andaluz]

? [�, es, juan, europeo, ?].

! No

? [todo, andaluz, es, europeo].

? [�, es, juan, europeo, ?].

! [juan, es, europeo]

? [�, qui�en, es, europeo, ?].

! [juan, es, europeo]

? muestra_reglas.

! [todo, andaluz, es, europeo]

! [juan, es, andaluz]

? fin.

Yes

u

De�nici�on

consulta(Base_de_reglas) :-

pregunta_y_lee(Entrada),

procesa_entrada(Entrada,Base_de_reglas).

pregunta_y_lee(Entrada) :-

write('? '),

read(Entrada).

Razonamiento con lenguaje natural

procesa_entrada(fin,_Base_de_reglas) :- !.

procesa_entrada(muestra_reglas,Base_de_reglas) :- !,

muestra_reglas(Base_de_reglas),

consulta(Base_de_reglas).

procesa_entrada(Oraci�on,Base_de_reglas) :-

phrase(oraci�on(Regla),Oraci�on), !,

consulta([Regla|Base_de_reglas]).

procesa_entrada(Pregunta,Base_de_reglas) :-

phrase(pregunta(P),Pregunta),

prueba(P,Base_de_reglas), !,

transforma(P,Clausula),

phrase(oraci�on(Clausula),Respuesta),

muestra_respuesta(Respuesta),

consulta(Base_de_reglas).

procesa_entrada(_Pregunta,Base_de_reglas) :-

muestra_respuesta('No'),

consulta(Base_de_reglas).

muestra_reglas([]).

muestra_reglas([Regla|Reglas]) :-

phrase(oraci�on(Regla),Oraci�on),

muestra_respuesta(Oraci�on),

muestra_reglas(Reglas).

muestra_respuesta(Respuesta) :-

write('! '),

write(Respuesta),

nl.

Razonamiento con lenguaje natural

prueba(true,_Base_de_reglas) :- !.

prueba((A,B),Base_de_reglas) :- !,

prueba(A,Base_de_reglas),

prueba(B,Base_de_reglas).

prueba(A,Base_de_reglas) :-

busca_clausula((A:-B),Base_de_reglas),

prueba(B,Base_de_reglas).

busca_clausula(Clausula,[Regla|_Reglas]) :-

copy_term(Regla,Clausula).

busca_clausula(Clausula,[_Regla|Reglas]) :-

busca_clausula(Clausula,Reglas).

transforma((A,B),[(A:-true)|Resto]) :- !,

transforma(B,Resto).

transforma(A,(A:-true)).

